
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 7.3.6.1, which is part of the LTE test suite iwd-EUTRA-TVB2009-12_D10wk22. 

2 Table of Contents

21
Overview

2
Table of Contents
2
3
Corrections required for LTE PDCP test case 7.3.6.1
3
3.1
Change 1
3



3 Corrections required for LTE PDCP test case 7.3.6.1
3.1 Change 1

	Testcase name
	f_TC_7_3_6_1_EUTRA

	Reason for change
	The test case expects PDCP PDUs in steps 7 and 8 to be received in separate PDUs, but not in a list. The current implementation does handle the PDUs in a list.  

	Summary of change
	The transmission of the PDCP PDUs to be looped back is scheduled with 100 ms in between, in order to ensure that the looped back PDCP PDUs will be received individually and not possibly in a list.
To achieve this, function UPSend is provided with a TimingInfo parameter. 

	TTCN module
	PDCP_Testcases.ttcn

	MCC160 Comments
	


Before:

	 function f_EUTRA_7_3_6_1_TestBody(inout PDCP_SS_State_Type p_PDCP_Rec)
  runs on EUTRA_PTC
{

…

 // Step 5: SS sends a PDCP Data PDU via RLC_UM RB with the following
  // content to the UE: D/C field = 1 (PDCP Data PDU) and PDCP SN = k
  // After having sent a PDU, the SS set Next PDCP TX SN = k+1
  // Exception: step 5 shall be repeated for k=3 to 4 (increment 1)
  f_UPSend(p_PDCP_Rec, eutra_Cell1, crs_IPv4_AnyIcmpEchoReply_4 ( px_IPv4_Address,
                                                                  px_IPv4_Address ));
  f_UPSend(p_PDCP_Rec, eutra_Cell1, crs_IPv4_AnyIcmpEchoReply_5 ( px_IPv4_Address,
                                                                  px_IPv4_Address ));
  // Step 6: SS resumes normal UL grant allocation
  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1, cs_TimingInfo_Now );
  // Step 7: Check: Does UE transmit a PDCP Data PDU with PDCP SN=3?
  p_PDCP_Rec.Next_PDCP_RX_SN := 3;
  f_UPRcv (p_PDCP_Rec, eutra_Cell1, crs_IPv4_AnyIcmpEchoReply_4 ( px_IPv4_Address,
                                                                  px_IPv4_Address ));
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7");
  // Step 8: Check: Does UE transmit a PDCP Data PDU with PDCP SN=4?
  f_UPRcv (p_PDCP_Rec, eutra_Cell1, crs_IPv4_AnyIcmpEchoReply_5 ( px_IPv4_Address,
                                                                  px_IPv4_Address ));
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");
} // f_EUTRA_7_3_6_1_TestBody



	function f_UPSend(inout PDCP_SS_State_Type p_PDCP_Rec,
                  CellId_Type p_CellId,
                  template (present) octetstring p_PdcpSdu)
  runs on EUTRA_PTC
{
  var template (value) PDCP_PDU_Type v_Pdu;
  var octetstring v_CipheredSdu;
  v_CipheredSdu := fx_AsCiphering (valueof (p_PdcpSdu),
                                   p_PDCP_Rec.CipheringAlgorithm,
                                   p_PDCP_Rec.KUPenc,
                                   int2oct(p_PDCP_Rec.Next_PDCP_TX_SN, 4),
                                   int2bit(p_PDCP_Rec.RB_Id.Drb - 1, 5));   /* R5-101050: BearerId changed to 5 bits */
  // for the time being TX_HFN will always be 0, so Next_PDCP_TX_SN is used instead of TX_COUNT !!!
  // Build PDCP pdu to be sent
  if ( p_PDCP_Rec.PDCP_SN_Size == PDCP_SNLength12) {
    v_Pdu := cs_PDCP_DataLongSN (p_PDCP_Rec.Next_PDCP_TX_SN, v_CipheredSdu);
  } else if ( p_PDCP_Rec.PDCP_SN_Size == PDCP_SNLength7){
    v_Pdu := cs_PDCP_DataShortSN (p_PDCP_Rec.Next_PDCP_TX_SN, v_CipheredSdu);
  } else {
    FatalError(__FILE__, __LINE__, "PDCP_SN_Size incorrect");
  }
  DRB.send (cas_DRB_PDCP_PDU_Req(p_CellId,
                                 cs_TimingInfo_Now,
                                 p_PDCP_Rec.RB_Id,
                                 v_Pdu));
  if (p_PDCP_Rec.Next_PDCP_TX_SN == p_PDCP_Rec.Maximum_PDCP_SN)
  {
    p_PDCP_Rec.Next_PDCP_TX_SN := 0;
    p_PDCP_Rec.TX_HFN := p_PDCP_Rec.TX_HFN + 1; // currently not used further
  }
  else
  {
    p_PDCP_Rec.Next_PDCP_TX_SN := p_PDCP_Rec.Next_PDCP_TX_SN + 1;
  }
} // f_UPSend



After:

	 function f_EUTRA_7_3_6_1_TestBody(inout PDCP_SS_State_Type p_PDCP_Rec)
  runs on EUTRA_PTC
{

…

 // Step 5: SS sends a PDCP Data PDU via RLC_UM RB with the following
  // content to the UE: D/C field = 1 (PDCP Data PDU) and PDCP SN = k
  // After having sent a PDU, the SS set Next PDCP TX SN = k+1
  // Exception: step 5 shall be repeated for k=3 to 4 (increment 1)
  // Both PDCP PDUs are scheduled with 100 ms in between to ensure they are looped back individually 
  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1); //100 ms in future
  f_UPSend(p_PDCP_Rec, eutra_Cell1, crs_IPv4_AnyIcmpEchoReply_4 ( px_IPv4_Address,
                                                                  px_IPv4_Address ),
                                                                  cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));
  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 100);
  f_UPSend(p_PDCP_Rec, eutra_Cell1, crs_IPv4_AnyIcmpEchoReply_5 ( px_IPv4_Address,
                                                                  px_IPv4_Address ),
                                                                  cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));
  // Step 6: SS resumes normal UL grant allocation
  // according to 36.523-3 this will be in less than 80 ms
  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1, cs_TimingInfo_Now );
  // Step 7: Check: Does UE transmit a PDCP Data PDU with PDCP SN=3?
  p_PDCP_Rec.Next_PDCP_RX_SN := 3;
  f_UPRcv (p_PDCP_Rec, eutra_Cell1, crs_IPv4_AnyIcmpEchoReply_4 ( px_IPv4_Address,
                                                                  px_IPv4_Address ));
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7");
  // Step 8: Check: Does UE transmit a PDCP Data PDU with PDCP SN=4?
  f_UPRcv (p_PDCP_Rec, eutra_Cell1, crs_IPv4_AnyIcmpEchoReply_5 ( px_IPv4_Address,
                                                                  px_IPv4_Address ));
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");
} // f_EUTRA_7_3_6_1_TestBody



	function f_UPSend(inout PDCP_SS_State_Type p_PDCP_Rec,
                  CellId_Type p_CellId,
                  template (present) octetstring p_PdcpSdu,
                  template (omit) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now)
  runs on EUTRA_PTC
{
  var template (value) PDCP_PDU_Type v_Pdu;
  var octetstring v_CipheredSdu;
  v_CipheredSdu := fx_AsCiphering (valueof (p_PdcpSdu),
                                   p_PDCP_Rec.CipheringAlgorithm,
                                   p_PDCP_Rec.KUPenc,
                                   int2oct(p_PDCP_Rec.Next_PDCP_TX_SN, 4),
                                   int2bit(p_PDCP_Rec.RB_Id.Drb - 1, 5));   /* R5-101050: BearerId changed to 5 bits */
  // for the time being TX_HFN will always be 0, so Next_PDCP_TX_SN is used instead of TX_COUNT !!!
  // Build PDCP pdu to be sent
  if ( p_PDCP_Rec.PDCP_SN_Size == PDCP_SNLength12) {
    v_Pdu := cs_PDCP_DataLongSN (p_PDCP_Rec.Next_PDCP_TX_SN, v_CipheredSdu);
  } else if ( p_PDCP_Rec.PDCP_SN_Size == PDCP_SNLength7){
    v_Pdu := cs_PDCP_DataShortSN (p_PDCP_Rec.Next_PDCP_TX_SN, v_CipheredSdu);
  } else {
    FatalError(__FILE__, __LINE__, "PDCP_SN_Size incorrect");
  }
  DRB.send (cas_DRB_PDCP_PDU_Req(p_CellId,
                                 p_TimingInfo,
                                 p_PDCP_Rec.RB_Id,
                                 v_Pdu));
  if (p_PDCP_Rec.Next_PDCP_TX_SN == p_PDCP_Rec.Maximum_PDCP_SN)
  {
    p_PDCP_Rec.Next_PDCP_TX_SN := 0;
    p_PDCP_Rec.TX_HFN := p_PDCP_Rec.TX_HFN + 1; // currently not used further
  }
  else
  {
    p_PDCP_Rec.Next_PDCP_TX_SN := p_PDCP_Rec.Next_PDCP_TX_SN + 1;
  }
} // f_UPSend

