
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of EMM test case 9.2.3.1.2 which are part of the LTE test suite.
The test case can be demonstrated to run with two LTE UEs (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_9_2_3_1_2
Test Group:
EMM
ATS Version:
iwd-EUTRA-TVB2009-03_D10wk22
System Simulator used:
Anite Conformance Toolset
UE(s) used:
Samsung Tanum UE, LG L2000 UE
Verification Status:
PASS
4 Corrections required for test case 9.2.3.1.2
4.1 Introduction

This section describes the changes required to make test case 9.2.3.1.2 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D10wk22 release.

4.2 Change 1

	Testcase name
	9.2.3.1.2

	Reason for change
	1) UE in the EMM Connected state, UE will not trigger Service Request procedure at step 2. So UE should be kept in closed Test Loop Mode B.
2) Old GUTI of UE should be retrieved from Cell A for TAU procedure as Cell A does not use the default TAI.

3) Send IP Packet and send RRC Connection Release after one second to Initiate Service Request procedure from UE (step2). 

4) EPS_UpdateTypeValue for TAU procedure should be tsc_EpsUpdate_TaUpdate.

5) Send RRC Connection Release on Cell A to initiate UE camp on Cell B after step4.

6) Wrong variable v_NasInd used to get the epsBearerContextStatus.

7) At step 7 and 8 is a redundant sequence as both UE and SS will have the EPS security context. Security Mode command can be sent to the UE only in case UE has a security context which is not yet in use. Here UE has got only 1 security context and that is already in use for sending Tracking area Update request message.
8) Because of change on step1 no dedicated bearer will be configured. So instead of checking EPS dedicated Active, to validate TP2: NAS signalling active? SS asks for UE identity Request. And receive UE identity response.
Note: Prose CR will be raised for these issues at the next RAN5 meeting.

	Summary of change
	1) Configured UE as in state Loopback Activated (state 4) with condition UE TEST LOOP MODE B on Cell A.
2) Initialised  v_GutiParams with f_EUTRA_CellInfo_GetGuti(eutra_CellA)
3) At step 1 removed the AT commands and introduced SS to transmit IP data on DRB1 and release RRC Connectio after one second.
4) At step 5 replaced v_UpdateType with tsc_EpsUpdate_TaUpdate.
5) Added RRC Connection release after 300 ms after step4.

6) Replaced v_NasInd with v_ReceivedAsp.Signalling.Nas[0].
7) Removed step 7 and 8.

8) Replace the step12 with UE identity request and response procedure.

	Source of change
	File : 


Before:
	function f_TC_9_2_3_1_2_EUTRA() runs on EUTRA_PTC

  {

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, LTEsinglecell);

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_CellConfig_SRBs_Only(eutra_CellB);

    // Preamble:

    //   - The UE is brought to state Registered, Idle Mode (state 2) on Cell A

    // ==========================================================================

    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

    f_EUTRA_GenericRbEst_Common(eutra_CellA,2,0,omit,omit,omit,omit,omit);
    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_3_1_2_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellB, E2_CONNECTED);

    // Now remove cell A

    f_EUTRA_ReleaseAllCells( );

  } // function f_TC_9_2_3_1_2_EUTRA

  function fl_TC_9_2_3_1_2_Body() runs on EUTRA_PTC

  {

    // ==========================================================================

    // initialize component specific data structures

    // ==========================================================================

    // get all data which are required for a GUTI

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellB);

    var template (value) TrackingAreaIdList v_TaiList;

    // ==========================================================================

    var EPS_BearerContextStatus v_EpsBearerCtxtStatus;

    var SRB_COMMON_IND v_ReceivedAsp;

    var template (value) CellPowerList_Type v_CellPowerList;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_AttDetValue_Type v_UpdateType := f_GetEAttachType();

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    timer t_Wait;

    //+ Step 1

    f_UT_RequestMODedicatedBearer (UT,

            hex2int(tsc_EpsDedicatedBearerId),

            hex2int(tsc_EpsDefaultBearerId),

            cs_508_EPS_QoS_Dedicated_1, cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId));
    //+ Step 2

    //+   The UE transmits SERVICE REQUEST message. --> SERVICE REQUEST

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                             tsc_RRC_TI_Def,

                                             cr_EstablishmentCause_MoData,

                                             cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                               cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    //+ Step 3 (specific msg. content)

    // *** no action required ***

    //+ Step 4

    //+   Set the cell type of cell A to the "Non-suitable cell".

    //+   Set the cell type of cell B to the "Serving cell".

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE ),

      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

    //+ Step 5 (specific msg. content)

    //+   Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message with

    //+   'Active' flag  before expiring T3417 timer (5 seconds)?? --> TRACKING AREA UPDATE REQUEST [1,P]

    t_Wait.start(f_EUTRA_SetTimerToleranceMax ( eutra_CellB, 5.0 ));

    alt {

    []  SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellB, cr_RRCConnectionRequest(?)))

      {

          f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellB, cs_TimingInfo_Now, tsc_RRC_TI_Def);

          SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellB,

                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                               cdr_TAU_Request_OldGuti_BearerEst(v_UpdateType,

                                                                                                 f_EUTRA_SecurityKSIasme_Get(),

                                                                                                 v_AdditionalUpdateType,

                                                                                                 f_GutiParameters2MobileIdentity(omit, v_GutiParams)))))

           -> value v_ReceivedAsp;

          t_Wait.stop;

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

      }

    []  t_Wait.timeout

      {

         f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 5 ");

      }

  }

    v_EpsBearerCtxtStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;
    //+ Step 7

    //+   The SS starts a NAS security mode command procedure to perform NAS

    //+   integrity protection. <-- SECURITY MODE COMMAND

    //+ Step 8

    //+   The UE responds to the NAS security mode command procedure. --> SECURITY

    //+   MODE COMPLETE

     f_EUTRA_NAS_AS_Security (eutra_CellB, tsc_SRB1);
    //+ Step 9

    //+   The SS transmits a TRACKING AREA UPDATE ACCEPT message. <-- TRACKING

    //+   AREA UPDATE ACCEPT

    v_TaiList := cds_TAIListNonConsecutive_tlv(

               f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity),

               { bit2oct(f_EUTRA_CellInfo_GetTAC(eutra_CellB)) });

    SRB.send(cas_SRB1_NasPdu_REQ(

                                 eutra_CellB,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_TAU_Accept(v_UpdateType,

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams),

                                                                  v_TaiList,

                                                                  v_EpsBearerCtxtStatus,

                                                                  f_EUTRA_GetCellLAI(eutra_CellA),

                                                                  omit,

                                                                  f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    //+ Step 10

    //+   Check: Does the UE transmit a TRACKING AREA UPDATE COMPLETE message? -->

    //+   TRACKING AREA UPDATE COMPLETE [1,P]

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete)));

    //+ Step 11

    //+   The SS waits 10seconds (T3440).

    f_Delay(10.0); // signalling connection release timer

    //+ Step 12

    //+   Check: does the test result of generic procedure indicate that a dedicated EPS bearer context is active? (TS36.508 clause 6.4.2.6) [2,P]

     f_EUTRA_508Check_DedicatedEPSIsActive(eutra_CellB, tsc_EpsDedicatedBearerId, cs_508_EPS_QoS_Dedicated_1, cds_Tft_Dedicated_2_Modify(tsc_EpsDedicatedBearerId));
  } // function fl_TC_9_2_3_1_2_Body


After:
	 function f_TC_9_2_3_1_2_EUTRA() runs on EUTRA_PTC

  {

    /// @desc Normal tracking area update / accepted / 'Active' flag set

    // Purpose

    // -------

    /* (1)

    with { UE in state EMM-SERVICE-REQUEST-INITIATED and EMM-CONNECTED}

    ensure that {

      when { UE detects entering a new tracking area already not included in the TAI list }

       then { UE sends TRACKING AREA UPDATE REQUEST message with ''Active'' flag before expiring T3417}

                 }

    (2)

    with { UE in state EMM-REGISTERED and EMM-CONNECTED}

    ensure that {

      when { UE sends TRACKING AREA UPDATE COMPLETE to NW }

       then { UE keeps the user plane and  the NAS signalling connection }

         }

    */

    // Pre-test conditions

    // -------------------

    // System Simulator:

    //   - Cell A (TAI-1:MCC1/MNC1/TAC1) and Cell B(TAI-2:MCC1/MNC1/TAC2) (FDD for

    //     Cell A and Cell B, or TDD for Cell A and Cell B).

    //   - Cell A is set to the "Serving cell" ,Cell B is set to the "Suitable neighbour cell".

    // UE:

    //   none

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, LTEsinglecell);

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_CellConfig_SRBs_Only(eutra_CellB);

    // Preamble:

    //   - The UE is brought to state Registered, Idle Mode (state 2) on Cell A

    // ==========================================================================
    f_EUTRA_Preamble(eutra_CellA, STATE2A_TESTLOOP_ModeB);

    f_EUTRA_RbEst_Def(eutra_CellA);

    f_EUTRA_CloseUE_TestLoopModeB(eutra_CellA, '0A'O);
    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_3_1_2_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellB, E2_CONNECTED);

    // Now remove cell A

    f_EUTRA_ReleaseAllCells( );

  } // function f_TC_9_2_3_1_2_EUTRA

  function fl_TC_9_2_3_1_2_Body() runs on EUTRA_PTC

  {

    // ==========================================================================

    // initialize constant test case data structures

    // ==========================================================================

    // const EPS_BearerIdentity v_EpsBearerId := '5'H;

    // (24.007 11.2.3.1.a)

    // const ProcedureTransactionIdentifier v_EpsPti := '01'O;

    // ==========================================================================

    // ==========================================================================

    // initialize component specific data structures

    // ==========================================================================

    // get all data which are required for a GUTI

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var template (value) TrackingAreaIdList v_TaiList;

    // ==========================================================================

    var EPS_BearerContextStatus v_EpsBearerCtxtStatus;

    var SRB_COMMON_IND v_ReceivedAsp;

    var template (value) CellPowerList_Type v_CellPowerList;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_AttDetValue_Type v_UpdateType := f_GetEAttachType();

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    timer t_Wait;

    var octetstring v_IPData :=  crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_RemoteAddress, px_IPv4_RemoteAddress );
    //+ Step 1   

    DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU(eutra_CellA, tsc_RbId_DRB1, cs_TimingInfo_Now, {v_IPData})); 
    f_Delay (1.0);

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA, tsc_RRC_TI_Def, other);
    //+ Step 2

    //+   The UE transmits SERVICE REQUEST message. --> SERVICE REQUEST

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                             tsc_RRC_TI_Def,

                                             cr_EstablishmentCause_MoData,

                                             cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                               cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    //+ Step 3 (specific msg. content)

    //+   The SS does not perform a radio bearer establishment procedure.

    //+ Note: SS does not send any AS or NAS messages to UE.

    // *** no action required ***

    // +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

    // The following messages are sent and shall be received on Cell B.

    // +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

    //+ Step 4

    //+   Set the cell type of cell A to the "Non-suitable cell".

    //+   Set the cell type of cell B to the "Serving cell".

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE ),

      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

    f_Delay (0.3);// Delay required to ensure cell power is effective before RRC Connection release is sent.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA, tsc_RRC_TI_Def, other);
    //+ Step 5 (specific msg. content)

    //+   Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message with

    //+   'Active' flag  before expiring T3417 timer (5 seconds)?? --> TRACKING AREA UPDATE REQUEST [1,P]

    t_Wait.start(f_EUTRA_SetTimerToleranceMax ( eutra_CellB, 5.0 ));

    alt {

    []  SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellB, cr_RRCConnectionRequest(?)))

      {

          f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellB, cs_TimingInfo_Now, tsc_RRC_TI_Def);

          SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellB,

                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                               cdr_TAU_Request_OldGuti_BearerEst(tsc_EpsUpdate_TaUpdate,
                                                                                                 f_EUTRA_SecurityKSIasme_Get(),

                                                                                                 v_AdditionalUpdateType,

                                                                                                 f_GutiParameters2MobileIdentity(omit, v_GutiParams)))))

           -> value v_ReceivedAsp;

          t_Wait.stop;

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

      }

    []  t_Wait.timeout

      {

         f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 5 ");

      }

  }

    v_EpsBearerCtxtStatus := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;
    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellB);
    //+ Step 9

    //+   The SS transmits a TRACKING AREA UPDATE ACCEPT message. <-- TRACKING

    //+   AREA UPDATE ACCEPT

    v_TaiList := cds_TAIListNonConsecutive_tlv(

               f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity),

               { bit2oct(f_EUTRA_CellInfo_GetTAC(eutra_CellB)) });

    SRB.send(cas_SRB1_NasPdu_REQ(

                                 eutra_CellB,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_TAU_Accept(v_UpdateType,

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams),

                                                                  v_TaiList,

                                                                  v_EpsBearerCtxtStatus,

                                                                  f_EUTRA_GetCellLAI(eutra_CellA),

                                                                  omit,

                                                                  f_GetAdditionalUpdateResult(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    //+ Step 10

    //+   Check: Does the UE transmit a TRACKING AREA UPDATE COMPLETE message? -->

    //+   TRACKING AREA UPDATE COMPLETE [1,P]

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete)));

    //+ Step 11

    //+   The SS waits 10seconds (T3440).

    f_Delay(10.0); // signalling connection release timer

    //+ Step 12

    //+   Check: does the test result of generic procedure indicate that a dedicated EPS bearer context is active? (TS36.508 clause 6.4.2.6) [2,P]     

    //Identity request and response.

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_IDENTITY_REQUEST(tsc_IdType_IMSI))));

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_IDENTITY_RESPONSE(f_Imsi2MobileIdentity(px_IMSI_Def)))));
  } // function fl_TC_9_2_3_1_2_Body


5 Execution Log Files
5.1 Samsung Tanum UE

The Samsung Tanum UE passed these test cases on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files tc_9_2_3_1_2_Samsung-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

5.2 LG L2000 UE
The LG L2000 UE passed these test cases on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files tc_9_2_3_1_2_LG-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

	[1]
	R5s100395: This archive comprises html format execution log file.

