
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of several test cases, which are part of the LTE/SAE test suite iwd-EUTRA-TVB2009-12_D10wk22. 

3 Corrections required for LTE/SAE iwd_10wk22 ATS
3.1 Change 1

	Testcase name
	f_TC_7_3_1_2_EUTRA

	Reason for change
	The test case is testing PDCP with 7 bit sequence number. Hence a 7 bit PDCP needs to be configured.
(Note: This change is linked with changes 1 and 2 below.)

	Summary of change
	Use the new function f_EUTRA_CellConfig_DefPlusUM_PDCPShortSN.

	TTCN module
	PDCP_Testcases.ttcn

	MCC-STF160
	


Before:

	function f_TC_7_3_1_2_EUTRA() runs on EUTRA_PTC

{ /* @desc Maintenance of PDCP sequence numbers (user plane, RLC UM, short PDCP SN (7 bits)) */

  // @sic R5-100082 sic@

  var PDCP_SS_State_Type v_PDCP_Rec;

  var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

  // Initialization, generic setup, DRB loopback

  f_EUTRA_Init(c1);

  fl_InitPDCP_Record(v_PDCP_Rec, UM_Mode, PDCP_SNLength7);

  // get FDD_TDD_Info

  v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell1);

  // Cell configuration, with one UM DRB at the SS side

  f_EUTRA_CellConfig_DefPlusUM(eutra_Cell1);
  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

  //set null AS ciphering algorithm

  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

  //entering UE test state state 3 with test case specific RLC parameters, with BSR and PHR disabled

  f_EUTRA_GenericRbEst_UM_Params (eutra_Cell1,

                                  cds_PDCP_Config_DRB_UM_SN5,

                                  cs_508_DRB_UL_UM_RLC,

                                  cs_508_DRB_DL_UM_RLC);

  // reconfigure DRB locally to set the required PDCP test mode

  f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB( eutra_Cell1, cs_DRB2_ConfigUM_PDCP_Mode(v_PDCP_Rec.PDCP_SN_Size) );

  // closing the test loop (mode A), entering state 4

  f_EUTRA_CloseUE_TestLoopModeA( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling );

  f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@

  f_EUTRA_7_3_1_2_TestBody (v_PDCP_Rec);

  f_EUTRA_TestBody_Set(false);

  // leaving UE test state 4

  //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

  //postamble

  f_EUTRA_Postamble (eutra_Cell1, E3_TESTMODE);

} // TC_7_3_1_2_EUTRA


After:

	function f_TC_7_3_1_2_EUTRA() runs on EUTRA_PTC

{ /* @desc Maintenance of PDCP sequence numbers (user plane, RLC UM, short PDCP SN (7 bits)) */

  // @sic R5-100082 sic@

  var PDCP_SS_State_Type v_PDCP_Rec;

  var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

  // Initialization, generic setup, DRB loopback

  f_EUTRA_Init(c1);

  fl_InitPDCP_Record(v_PDCP_Rec, UM_Mode, PDCP_SNLength7);

  // get FDD_TDD_Info

  v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell1);

  // Cell configuration, with one UM DRB at the SS side

  f_EUTRA_CellConfig_DefPlusUM_PDCPShortSN(eutra_Cell1); 
  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

  //set null AS ciphering algorithm

  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

  //entering UE test state state 3 with test case specific RLC parameters, with BSR and PHR disabled

  f_EUTRA_GenericRbEst_UM_Params (eutra_Cell1,

                                  cds_PDCP_Config_DRB_UM_SN5,

                                  cs_508_DRB_UL_UM_RLC,

                                  cs_508_DRB_DL_UM_RLC);

  // reconfigure DRB locally to set the required PDCP test mode

  f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB( eutra_Cell1, cs_DRB2_ConfigUM_PDCP_Mode(v_PDCP_Rec.PDCP_SN_Size) );

  // closing the test loop (mode A), entering state 4

  f_EUTRA_CloseUE_TestLoopModeA( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling );

  f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@

  f_EUTRA_7_3_1_2_TestBody (v_PDCP_Rec);

  f_EUTRA_TestBody_Set(false);

  // leaving UE test state 4

  //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

  //postamble

  f_EUTRA_Postamble (eutra_Cell1, E3_TESTMODE);

} // TC_7_3_1_2_EUTRA


3.2 Change 2

	Testcase name
	f_EUTRA_CellConfig_DefPlusUM_PDCPShortSN

	Reason for change
	A new CellConfig function is needed for configuring PDCP with short SN. 
(Note: This change is linked to change 1 above and change 3 below) 

	Summary of change
	New function coded

	TTCN module
	EUTRA_ConfigurationSteps.ttcn

	MCC-STF160
	


Before:

	


After: 

	function f_EUTRA_CellConfig_DefPlusUM_PDCPShortSN (CellId_Type p_CellId) runs on EUTRA_PTC

  {

    f_EUTRA_CellConfig_Def ( p_CellId );

    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(p_CellId, 

            cs_OneDRB_ConfigUM(tsc_DRB2, cs_508_PDCP_SN7_Config_DRB_UM));

  };

 


3.3 Change 3

	Testcase name
	cs_508_PDCP_SN7_Config_DRB_UM

	Reason for change
	New  template to configure PDCP with 7 bits is needed.
(Note: This change is linked to change 1 and 2 above)

	Summary of change
	New function coded

	TTCN module
	EUTRA_DRB_Templates.ttcn

	MCC-STF160
	


Before:

After:

template (value) PDCP_Config cs_508_PDCP_SN7_Config_DRB_UM :=

  { 

    discardTimer := ms100,

    rlc_AM := omit,

    rlc_UM := {

      pdcp_SN_Size := len7bits 

    },

    headerCompression := {

      notUsed := NULL

    }

  };
3.4 Change 4

	Function name
	fl_TC_9_3_1_1_Body

	Reason for change
	The IP packet sent in step 1 will be looped back and needs to be handled. Note that the prose specification has been changed in RAN5#47 (CR R5-103696) and the re-establishment is now not rejected but completed.
(Note: This change is related to change 5 below)

	Summary of change
	A default handler for handling the uplink IP packet is activated  prior to DRB establishment. 

	Source of change
	NAS_ServiceRequest.ttcn

	MCC-STF160
	


Before:

	function fl_TC_9_3_1_1_Body() runs on EUTRA_PTC

{

  var NAS_MSG_Indication_Type v_NasInd;

  //Step 1. The SS transmits one IP Packet to the UE (over default DRB 0)

  DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU(eutra_CellA, tsc_RbId_DRB1, cs_TimingInfo_Now, {crs_IPv4_AnyIcmpEchoReply_1(tsc_IPv4RemoteAddress, px_IPv4_Address)})); // @sic R5s100135 sic@

  //+ Step 2 The SS waits 1 second after the IP packet has been transmitted in step 1

  //+ and then transmits an RRCConnectionRelease message

  f_Delay (1.0);

  f_EUTRA_RRC_ConnectionRelease(eutra_CellA, tsc_RRC_TI_Def, other);

  //+ Step 3

  //+   Check: Check: Does the UE transmit an RRCConnectionRequest message with establishmentCause set

  //             to mo-Data followed by a SERVICE REQUEST message? [1,P]

  //+ Note: Triggered when timer T_delay_modeB (IP PDU delay time) expires and

  //+ pending uplink data exist in buffered PDCP SDUs according to [25] clause

  //+ 5.4.4.3.

  v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                             tsc_RRC_TI_Def,

                                             cr_EstablishmentCause_MoData,

                                             cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                               cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

  // + Step 4-7: @sic R5-102180 sic@

  // Steps 6 to 9 of the generic radio bearer establishment procedure (TS 36.508 5.4.3.3-1) are executed

  // to successfully complete the service request procedure

  f_EUTRA_ServiceReqPostamble_508Steps6_9 (eutra_CellA, v_NasInd);

} // function fl_TC_9_3_1_1_Body


After:

	function fl_TC_9_3_1_1_Body() runs on EUTRA_PTC

{

  var NAS_MSG_Indication_Type v_NasInd;

  var default v_DrbDefault;
  //Step 1. The SS transmits one IP Packet to the UE (over default DRB 0)

  DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU(eutra_CellA, tsc_RbId_DRB1, cs_TimingInfo_Now, {crs_IPv4_AnyIcmpEchoReply_1(tsc_IPv4RemoteAddress, px_IPv4_Address)})); // @sic R5s100135 sic@

  //+ Step 2 The SS waits 1 second after the IP packet has been transmitted in step 1

  //+ and then transmits an RRCConnectionRelease message

  f_Delay (1.0);

  f_EUTRA_RRC_ConnectionRelease(eutra_CellA, tsc_RRC_TI_Def, other);

  //+ Step 3

  //+   Check: Check: Does the UE transmit an RRCConnectionRequest message with establishmentCause set

  //             to mo-Data followed by a SERVICE REQUEST message? [1,P]

  //+ Note: Triggered when timer T_delay_modeB (IP PDU delay time) expires and

  //+ pending uplink data exist in buffered PDCP SDUs according to [25] clause

  //+ 5.4.4.3.

  v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                             tsc_RRC_TI_Def,

                                             cr_EstablishmentCause_MoData,

                                             cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                               cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");
  // Set a handler for the uplink IP packet which will be sent after DRB establishment

  v_DrbDefault := activate( a_DrbDefault_9311());
  // + Step 4-7: @sic R5-102180 sic@

  // Steps 6 to 9 of the generic radio bearer establishment procedure (TS 36.508 5.4.3.3-1) are executed

  // to successfully complete the service request procedure

  f_EUTRA_ServiceReqPostamble_508Steps6_9 (eutra_CellA, v_NasInd);

} // function fl_TC_9_3_1_1_Body


3.5 Change 5

	New Default handler
	a_DrbDefault_9311()

	Reason for change
	see Change 4 above

	Summary of change
	New default handler for test 9.3.1.1 coded

	Source of change
	NAS_ServiceRequest.ttcn

	MCC-STF160
	


Before:

	-


After:

	  altstep a_DrbDefault_9311() runs on EUTRA_PTC

  {

   []DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_CellA,

                                                    tsc_RbId_DRB1,

                                                    ?,

                                                    ?))

      {

        repeat;

      };

  };  

}


3.6 Change 6
	Function name
	fl_EUTRA_MatchEutraCapMsgs 

	Reason for change
	The check on the number of elements in the interFreqEUTRA-BandList IE is not correct. 

	Summary of change
	The count of  the number of  number interRAT-BandList has been corrected.


An accompanying prose CR to TS 36.523-1 Table 8.5.4.1.3.3-2 will be raised at the next RAN5 meeting stating that the check to interRAT-BandList shall be "m entries (contents not checked), but m shall be equal to the number of bands listed in each IE present in interRAT-Parameters"

(Note: This change is related to change 7 and 8 below.)

	Source of change
	RRC_Others.ttcn

	MCC-STF160
	


Before:

	  function fl_EUTRA_MatchEutraCapMsgs ( octetstring p_Octetstring,

                                        template UE_EUTRA_Capability p_Exptected_Ue_Eutra_Capability ) runs on EUTRA_PTC

  {

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;

    var integer i, k, l;

    var B40_Type v_NrOfSuppEutraBandsBitstring := fl_GetNrOfSuppEutraBandsBitstring();

    var charstring v_FailedEutraBand;

    var boolean v_BandSuppMatch := false;

    v_ReceivedCapabilityMsg := f_EUTRA_DecodeEutraCapMsg ( p_Octetstring );

    //* @desc Check that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

    if ( lengthof ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA ) != fl_GetNrOfSuppEutraBands() ) {

      //* @verdict fail Number of Supported EUTRA Bands according to PICS parameters

      // does not match number of entries in SupportedBandListEUTRA

      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Missmatch between pc_eBanda_Supp and number of reported supported EUTRA bands");

    }

    //* @desc Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

    //* Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS (received from fl_GetNrOfSuppEutraBandsBitstring).

    for ( k := 0; k < lengthof ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA ); k := k + 1 ){

      v_BandSuppMatch := false;

      for ( l := 0; l < lengthof ( v_NrOfSuppEutraBandsBitstring ); l := l + 1 ){

        //* @desc For each supported band (according to the set PICS) match the band with what is reported according to k.

        if ( (l+1) == v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA ){

          if ( v_NrOfSuppEutraBandsBitstring[l] == '0'B ){

            v_FailedEutraBand := int2char ( l + 1 );

            //* @verdict fail PICS setting does not match reported capability for band nr. l+1

            // or band-support har been reported more than once

            f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, v_FailedEutraBand);

          }

          else {

            //* UE reports support of band v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA

            // and the correspponing PICS parameter is also set.

            v_NrOfSuppEutraBandsBitstring[l] := '0'B;

            v_BandSuppMatch := true;

          }

        }

      }

      if ( v_BandSuppMatch == false ) {

        //* @verdict fail UE reports support of band v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA

        // but corresponding PICS is not set to true.

        v_FailedEutraBand := int2char ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA );

        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, v_FailedEutraBand);

      }

    }

    //* @desc interFreqEUTRA-BandList SEQUENCE (SIZE (1..maxBands)) OF SEQUENCE {

    //* same number of entries like in supportedEUTRA-BandList

    //* Match number of entries in interFreqEUTRA-BandList with number of entries in supportedEUTRA-BandList

    for ( i := 0; i < lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA ); i := i + 1 ) {

      if ( lengthof ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA ) != lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interFreqBandList ) )

      {

        //* @verdict fail Number of interFreqEUTRA-BandList does not match nr. of supportedEUTRA-BandList

        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Number of interFreqEUTRA-BandList does not match nr. of supportedEUTRA-BandList");

      }

    }

    for ( i := 0; i < lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA ); i := i + 1 ) {

      if ( ispresent ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interRAT_BandList ) )

      {

        //* @desc Check that number of interRAT-Parameters matches interRAT-BandList

        if ( lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interRAT_BandList ) != 

              sizeof ( v_ReceivedCapabilityMsg.interRAT_Parameters ) )
        // Note: sizeof returns the number of fields being present in a record what is intended here

        {

          //* @verdict fail Number of elements in interRAT-Parameters does not match number of entires in interRAT-BandList

          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Number of elements in interRAT-Parameters does not match number of entires in interRAT-BandList");

        }

      }

    }

    //* @desc Match the received decoded template with the expected template

    if ( not match ( v_ReceivedCapabilityMsg, p_Exptected_Ue_Eutra_Capability )){

      //* @verdict fail Received decoded template does not match with the expected template

      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Received decoded template does not match with the expected template");

    }

  }




After:

	  function fl_EUTRA_MatchEutraCapMsgs ( octetstring p_Octetstring,

                                        template UE_EUTRA_Capability p_Exptected_Ue_Eutra_Capability ) runs on EUTRA_PTC

  {

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;

    var integer i, k, l;

    var B40_Type v_NrOfSuppEutraBandsBitstring := fl_GetNrOfSuppEutraBandsBitstring();

    var charstring v_FailedEutraBand;

    var boolean v_BandSuppMatch := false;

    v_ReceivedCapabilityMsg := f_EUTRA_DecodeEutraCapMsg ( p_Octetstring );

    //* @desc Check that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

    if ( lengthof ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA ) != fl_GetNrOfSuppEutraBands() ) {

      //* @verdict fail Number of Supported EUTRA Bands according to PICS parameters

      // does not match number of entries in SupportedBandListEUTRA

      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Missmatch between pc_eBanda_Supp and number of reported supported EUTRA bands");

    }

    //* @desc Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

    //* Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS (received from fl_GetNrOfSuppEutraBandsBitstring).

    for ( k := 0; k < lengthof ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA ); k := k + 1 ){

      v_BandSuppMatch := false;

      for ( l := 0; l < lengthof ( v_NrOfSuppEutraBandsBitstring ); l := l + 1 ){

        //* @desc For each supported band (according to the set PICS) match the band with what is reported according to k.

        if ( (l+1) == v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA ){

          if ( v_NrOfSuppEutraBandsBitstring[l] == '0'B ){

            v_FailedEutraBand := int2char ( l + 1 );

            //* @verdict fail PICS setting does not match reported capability for band nr. l+1

            // or band-support har been reported more than once

            f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, v_FailedEutraBand);

          }

          else {

            //* UE reports support of band v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA

            // and the correspponing PICS parameter is also set.

            v_NrOfSuppEutraBandsBitstring[l] := '0'B;

            v_BandSuppMatch := true;

          }

        }

      }

      if ( v_BandSuppMatch == false ) {

        //* @verdict fail UE reports support of band v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA

        // but corresponding PICS is not set to true.

        v_FailedEutraBand := int2char ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA );

        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, v_FailedEutraBand);

      }

    }

    //* @desc interFreqEUTRA-BandList SEQUENCE (SIZE (1..maxBands)) OF SEQUENCE {

    //* same number of entries like in supportedEUTRA-BandList

    //* Match number of entries in interFreqEUTRA-BandList with number of entries in supportedEUTRA-BandList

    for ( i := 0; i < lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA ); i := i + 1 ) {

      if ( lengthof ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA ) != lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interFreqBandList ) )

      {

        //* @verdict fail Number of interFreqEUTRA-BandList does not match nr. of supportedEUTRA-BandList

        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Number of interFreqEUTRA-BandList does not match nr. of supportedEUTRA-BandList");

      }

    }

    for ( i := 0; i < lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA ); i := i + 1 ) {

      if ( ispresent ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interRAT_BandList ) )

      {

        //* @desc Check that number of interRAT-Parameters matches interRAT-BandList

        if ( lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interRAT_BandList ) != 

              fl_Count_interRAT_Parameters ( v_ReceivedCapabilityMsg ) )
        // Note: sizeof returns the number of fields being present in a record what is intended here

        {

          //* @verdict fail Number of elements in interRAT-Parameters does not match number of entires in interRAT-BandList

          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Number of elements in interRAT-Parameters does not match number of entires in interRAT-BandList");

        }

      }

    }

    //* @desc Match the received decoded template with the expected template

    if ( not match ( v_ReceivedCapabilityMsg, p_Exptected_Ue_Eutra_Capability )){

      //* @verdict fail Received decoded template does not match with the expected template

      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Received decoded template does not match with the expected template");

    }

  }


3.7 Change 7

	Function name
	fl_Count_interRAT_Parameters

	Reason for change
	New function needed for  counting the number of IRAT parameters listed in all IEs in interRAT_Parameters.
(Note: This change is related to change 6 above and change 8 below.)

	Summary of change
	New funtion coded

	Source of change
	RRC_Others.ttcn

	MC-STF160
	


Before:

	


After:

	/*

   * @desc      Function returns the number of IRAT parameters listed in all IEs in interRAT_Parameters.

   * @return    integer

   * @status    

   */

  function fl_Count_interRAT_Parameters ( UE_EUTRA_Capability p_UECaps ) return integer

  {

    var integer v_NiratPars := 0;

    if (ispresent (p_UECaps.interRAT_Parameters.utraFDD)){

      v_NiratPars := v_NiratPars + lengthof(p_UECaps.interRAT_Parameters.utraFDD.supportedBandListUTRA_FDD);

    };

    if (ispresent (p_UECaps.interRAT_Parameters.utraTDD128)){

      v_NiratPars := v_NiratPars + lengthof(p_UECaps.interRAT_Parameters.utraTDD128.supportedBandListUTRA_TDD128);

    };

    if (ispresent (p_UECaps.interRAT_Parameters.utraTDD384)){

      v_NiratPars := v_NiratPars + lengthof(p_UECaps.interRAT_Parameters.utraTDD384.supportedBandListUTRA_TDD384);

    };

    if (ispresent (p_UECaps.interRAT_Parameters.utraTDD768)){

      v_NiratPars := v_NiratPars + lengthof(p_UECaps.interRAT_Parameters.utraTDD768.supportedBandListUTRA_TDD768);

    };

    if (ispresent (p_UECaps.interRAT_Parameters.geran)){

      v_NiratPars := v_NiratPars + lengthof(p_UECaps.interRAT_Parameters.geran.supportedBandListGERAN);

    }

    if (ispresent (p_UECaps.interRAT_Parameters.cdma2000_HRPD)){

      v_NiratPars := v_NiratPars + lengthof(p_UECaps.interRAT_Parameters.cdma2000_HRPD.supportedBandListHRPD);

    }

    if (ispresent (p_UECaps.interRAT_Parameters.cdma2000_1xRTT)){

      v_NiratPars := v_NiratPars + lengthof(p_UECaps.interRAT_Parameters.cdma2000_1xRTT.supportedBandList1XRTT);

    }

    return v_NiratPars;

  }



3.8 Change 8

	Function name
	fl_EUTRA_SetExpectedUeEutraCap

	Reason for change
	Incorrect code for template generation.

(Note: This change is related to change 6 and 7 above)

	Summary of change
	See below

	Source of change
	RRC_Others.ttcn

	MCC-STF160
	


Before:

	  function fl_EUTRA_SetExpectedUeEutraCap ( template UE_EUTRA_Capability p_UE_EUTRA_Capability )

    return template UE_EUTRA_Capability

  {

    var template UE_EUTRA_Capability v_UeCapabilityTemplate := p_UE_EUTRA_Capability;

    if ( pc_FDD ) {

      v_UeCapabilityTemplate.interRAT_Parameters.utraFDD := ?;

    }

    if ( pc_TDD_LCR ){

      v_UeCapabilityTemplate.interRAT_Parameters.utraTDD128 := ?;

    }

    if ( pc_TDD_HCR ){

      v_UeCapabilityTemplate.interRAT_Parameters.utraTDD384 := ?;

    }

    if ( pc_TDD_VHCR ){

      v_UeCapabilityTemplate.interRAT_Parameters.utraTDD768 := ?;

    }

    if ( pc_GERAN ){

      v_UeCapabilityTemplate.interRAT_Parameters.geran := ?;

    }

    if ( pc_HRPD ){

      v_UeCapabilityTemplate.interRAT_Parameters.cdma2000_HRPD := ?;

    }

    if ( pc_1xRTT ){

      v_UeCapabilityTemplate.interRAT_Parameters.cdma2000_HRPD := ?;

    }
    return v_UeCapabilityTemplate;

  }




After:

	  function fl_EUTRA_SetExpectedUeEutraCap ( template UE_EUTRA_Capability p_UE_EUTRA_Capability )

    return template UE_EUTRA_Capability

  {

    var template UE_EUTRA_Capability v_UeCapabilityTemplate := p_UE_EUTRA_Capability;

    if ( pc_FDD ) {

      v_UeCapabilityTemplate.interRAT_Parameters.utraFDD := ?;

    }

    if ( pc_TDD_LCR ){

      v_UeCapabilityTemplate.interRAT_Parameters.utraTDD128 := ?;

    }

    if ( pc_TDD_HCR ){

      v_UeCapabilityTemplate.interRAT_Parameters.utraTDD384 := ?;

    }

    if ( pc_TDD_VHCR ){

      v_UeCapabilityTemplate.interRAT_Parameters.utraTDD768 := ?;

    }

    if ( pc_GERAN ){

      v_UeCapabilityTemplate.interRAT_Parameters.geran := ?;

    }

    if ( pc_HRPD ){

      v_UeCapabilityTemplate.interRAT_Parameters.cdma2000_HRPD := ?;

    }

    if ( pc_1xRTT ){

      v_UeCapabilityTemplate.interRAT_Parameters.cdma2000_1xRTT := ?;

    }
    return v_UeCapabilityTemplate;

  }




3.9 Change 9

	Function name
	f_EUTRA_SS_SetupSchedulingInfo

	Reason for change
	RV values were not configured properly for SI scheduling when using DCI format 1C. The RV was always set to 0, but in case of DCI format 1C the RV depends on the Subframe Offset, as described in Section 5.3.1 of 36.321:

When the UE needs to read BCCH, the UE may, based on the scheduling information from RRC:

-
if a downlink assignment for this TTI has been received on the PDCCH for the SI-RNTI;
-
if the redundancy version is not defined in the PDCCH format:

-
the redundancy version of the received downlink assignment for this TTI is determined by RVK = ceiling(3/2*k) modulo 4, where k depends on the type of system information message: for SystemInformationBlockType1 message, k = (SFN/2) modulo 4, where SFN is the system frame number; for SystemInformation messages, k=i modulo 4, i =0,1,…, nsw–1, where i denotes the subframe number within the SI window nsw;
-
indicate a downlink assignment and redundancy version for the dedicated broadcast HARQ process to the HARQ entity for this TTI.

	Summary of change
	RV values for SI scheduling are added and overwritten for DCI format 1C according to the specified subframe offsets.

	Source of change
	EUTRA_ConfigurationSteps.ttcn

	MCC-STF160
	


Before:

	  function f_EUTRA_SS_SetupSchedulingInfo(  SystemInformationBlockType1 p_SIB1,

                                            EUTRA_FDD_TDD_Mode_Type  p_FDD_TDD )

                                            return template (value) AllSiSchedul_Type

  {

    var template (value) SiSchedulList_Type v_SiSchedulList;

    var integer v_NoOfSIs := lengthof(p_SIB1.schedulingInfoList);     // TTCN-3 3.4.1: sizeof replaced by lengthof

    var integer v_SubframeOffset;                          /* offset within the SI-window */

    var integer I;

    for (I := 0; I < v_NoOfSIs; I:= I + 1) {

      /* Assigne the subframe offset according to 36.523-3 Table 7.7.2.2;

         mapping of the SIBs to the SIs is specified in 36.508 cl. 4.4.3.1.2 "Scheduling of system information blocks"

         => there is a maximum of 4 SIs */

     select (p_FDD_TDD) {

        case (FDD) {

         select (I) {

           case (0) { v_SubframeOffset := 1; }  // Subframe offset for SI1

           case (1) { v_SubframeOffset := 1; }  // Subframe offset for SI2

                                                // R5-096691: 3 -> 1

           case (2) { v_SubframeOffset := 3; }  // Subframe offset for SI3

           case (3) { v_SubframeOffset := 7; }  // Subframe offset for SI4

           case else {  FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI"); }

          }

        }

        case (TDD) {  // R5-096691

         select (I) {

           case (0) { v_SubframeOffset := 4; }  // Subframe offset for SI1

           case (1) { v_SubframeOffset := 4; }  // Subframe offset for SI2

           case (2) { v_SubframeOffset := 9; }  // Subframe offset for SI3

           case (3) { v_SubframeOffset := 9; }  // Subframe offset for SI4

           case else {  FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI"); }

          }

        }

    }

    v_SiSchedulList[I] := cs_SiSchedul_Def(p_SIB1.schedulingInfoList[I].si_Periodicity, v_SubframeOffset);

  }

    return cs_SiSchedulList_Def(p_SIB1.si_WindowLength, v_SiSchedulList);

  };


After:

	  function f_EUTRA_SS_SetupSchedulingInfo(  SystemInformationBlockType1 p_SIB1,

                                            EUTRA_FDD_TDD_Mode_Type  p_FDD_TDD )

                                            return template (value) AllSiSchedul_Type

  {

    var template (value) SiSchedulList_Type v_SiSchedulList;

    var integer v_NoOfSIs := lengthof(p_SIB1.schedulingInfoList);     // TTCN-3 3.4.1: sizeof replaced by lengthof

    var integer v_SubframeOffset;                          /* offset within the SI-window */

    var integer v_RV := 0;
    var integer I;

    for (I := 0; I < v_NoOfSIs; I:= I + 1) {

      /* Assigne the subframe offset according to 36.523-3 Table 7.7.2.2;

         mapping of the SIBs to the SIs is specified in 36.508 cl. 4.4.3.1.2 "Scheduling of system information blocks"

         => there is a maximum of 4 SIs */

     select (p_FDD_TDD) {

        case (FDD) {

         select (I) {

           case (0) {

             v_SubframeOffset := 1;

             v_RV := 2;
           }  // Subframe offset for SI1

           case (1) {

             v_SubframeOffset := 1;

             v_RV := 2;
           }  // Subframe offset for SI2

              // R5-096691: 3 -> 1

           case (2) {

             v_SubframeOffset := 3;

             v_RV := 1;
           }  // Subframe offset for SI3

           case (3) {

             v_SubframeOffset := 7;

             v_RV := 3;
           }  // Subframe offset for SI4

           case else {  FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI"); }

          }

        }

        case (TDD) {  // R5-096691

         select (I) {

           case (0) {

             v_SubframeOffset := 4;

             v_RV := 2;
           }  // Subframe offset for SI1

           case (1) {

             v_SubframeOffset := 4;

             v_RV := 2;
           }  // Subframe offset for SI2

           case (2) {

             v_SubframeOffset := 9;

             v_RV := 2;
           }  // Subframe offset for SI3

           case (3) {

             v_SubframeOffset := 9;

             v_RV := 2;
           }  // Subframe offset for SI4

           case else {  FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI"); }

          }

        }

    }

    v_SiSchedulList[I] := cs_SiSchedul_Def(p_SIB1.schedulingInfoList[I].si_Periodicity, v_SubframeOffset);

    if (v_SiSchedulList[I].Window[0].DciInfo.Format == dci_1C) {

      v_SiSchedulList[I].Window[0].DciInfo.RedundancyVersionList[0] := v_RV;

    }
  }

    return cs_SiSchedulList_Def(p_SIB1.si_WindowLength, v_SiSchedulList);

  };


3.10 Change 10

	Function name
	cr_UDP_SOCKET_CNF

	Reason for change
	A UDP socket creation does not have any remote port info. Hence it can not report any.
(Note: This change is related to changes 11 and 12 below)

	Summary of change
	Change the template of the UDP_SOCKET_CNF to not allow omit for the Remote info. 

	Source of change
	IP_PTC_Templates.ttcn


Before:

	   template IP_SOCKET_IND cr_UDP_SOCKET_CNF(template IP_Socket_Type p_LocalSocket) :=

  {

    CTRL := {

      ConnectionId := cr_UDP_ConnectionId(p_LocalSocket, cr_IP_Socket_Any),

      Ind := {

        UDP := {

          SocketCnf := true

        }

      }

    }

  };


After:

	   template IP_SOCKET_IND cr_UDP_SOCKET_CNF(template IP_Socket_Type p_LocalSocket) :=

  {

    CTRL := {

      ConnectionId := cr_UDP_ConnectionId(p_LocalSocket, * /*cr_IP_Socket_Any*/),

      Ind := {

        UDP := {

          SocketCnf := true

        }

      }

    }

  };


3.11 Change 11

	New Default handler
	f_EUTRA_SS_SRBs_DRBs_Release

	Reason for change
	Wrong index integer used in the loop.

(Note: This change is related to change 10 above and change 12 below)

	Summary of change
	Exchanged i by k in loop.

	Source of change
	EUTRA_ConfigurationSteps.ttcn


Before:

	  function f_EUTRA_SS_SRBs_DRBs_Release ( CellId_Type  p_CellId,

                                          template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                          DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def ) runs on EUTRA_PTC

  {

    var template (value) RadioBearerList_Type v_SrbDrbList := {

      cs_SRB_Release(tsc_SRB0),

      cs_SRB_Release(tsc_SRB1),

      cs_SRB_Release(tsc_SRB2)

    };

    var integer i := lengthof(v_SrbDrbList);

    var integer k;

    for (k:=0; k < lengthof(p_DrbList); k:= k+1) {

      v_SrbDrbList[i+k] := cs_DRB_Release(p_DrbList[i]);

    }

    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, v_SrbDrbList, p_TimingInfo);

  }


After:

	  function f_EUTRA_SS_SRBs_DRBs_Release ( CellId_Type  p_CellId,

                                          template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                          DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def ) runs on EUTRA_PTC

  {

    var template (value) RadioBearerList_Type v_SrbDrbList := {

      cs_SRB_Release(tsc_SRB0),

      cs_SRB_Release(tsc_SRB1),

      cs_SRB_Release(tsc_SRB2)

    };

    var integer i := lengthof(v_SrbDrbList);

    var integer k;

    for (k:=0; k < lengthof(p_DrbList); k:= k+1) {

      v_SrbDrbList[i+k] := cs_DRB_Release(p_DrbList[k]); //@RSwk22 change i to k
    }

    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, v_SrbDrbList, p_TimingInfo);

  }


3.12 Change 12

	New Default handler
	f_EUTRA_SS_SRBs_DRBs_Config

	Reason for change
	Wrong index integer used in the loop.
(Note: This change is related to changes 10 and 11 above)

	Summary of change
	Exchanged i by k in loop.

	Source of change
	EUTRA_ConfigurationSteps.ttcn


Before:

	  function f_EUTRA_SS_SRBs_DRBs_Config ( CellId_Type  p_CellId,

                                         template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                         template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def ) runs on EUTRA_PTC

  {

    var template (value) RadioBearerList_Type v_SrbDrbList := {

      cs_SRB0_ConfigDef,

      cs_SRB1_ConfigDef,

      cs_SRB2_ConfigDef

    };

    var integer i := lengthof(v_SrbDrbList);

    var integer k;

    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {

      v_SrbDrbList[i+k] := p_DrbConfigList[i];

    }

    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, v_SrbDrbList, p_TimingInfo);

  }


After:

	  function f_EUTRA_SS_SRBs_DRBs_Config ( CellId_Type  p_CellId,

                                         template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                         template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def ) runs on EUTRA_PTC

  {

    var template (value) RadioBearerList_Type v_SrbDrbList := {

      cs_SRB0_ConfigDef,

      cs_SRB1_ConfigDef,

      cs_SRB2_ConfigDef

    };

    var integer i := lengthof(v_SrbDrbList);

    var integer k;

    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {

      v_SrbDrbList[i+k] := p_DrbConfigList[k]; //@RSwk22 change i to k
    }

    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, v_SrbDrbList, p_TimingInfo);

  }

