
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 12.2.2, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
tc_12_2_2
Test Group:
DRB
ATS Version:
iwd-EUTRA-TVB2009-12_D10wk22
System Simulator used:
Anite Conformance Toolset
UE used:
LG Electronics L2000 UE, Samsung Tanum UE
Verification Status:
PASS
4 Corrections required for test case 12.2.2
4.1 Introduction

This section describes the changes required to make test case 12.2.2 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-12_D10wk22 release.

4.2 Change 1

	Testcase name
	f_TC_12_2_2_EUTRA

	Reason for change
	1. This test case requires both AM and UM bearers to be configured. The drx_InactivityTimer of 100 ms is too short for the UM bearer as by default SS is configured to send TA once in 600 ms. As a result UE will go into DRX mode.
2. The PDCP discard timer for the UM bearer needs to be increased as in the test case SS gives the grant for the loop back data in UL after 10 Frames for step 5 of generic procedure 12.1.1.

Note: Prose CR will be raised for above two changes at the next RAN5 meeting.

	Summary of change
	1. A Boolean parameter p_TC_12_2_2 is added in the f_GenericProcedure_12_1_1 procedure to identify this test case so that special configuration can be done based on this parameter. A new function f_EUTRA_LoopBackActivation_State4_Generic_1222 is added which is called in f_GenericProcedure_12_1_1 to do the special configuration.
2. The PDCP discard timer is increased to 300ms. 

	Source of change
	Files : DRB_12.ttcn, EUTRA_CommonProcedures.ttcn, EUTRA_NASSteps.ttcn, EUTRA_DRB_Templates.ttcn


Before:

	  function f_TC_12_2_2_EUTRA ( ) runs on EUTRA_PTC

  { /* DRB/12.2

       12.2.2 : Data transfer of E-UTRA radio bearer combinations 2,4,7 and 10

    */

    ///////////////////////////////////////////////////////////////////////////////////

    //      PREAMBLE

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_Init (c1);

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    // Bring UE to state 3

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true);

    // Test body starts

    // E-UTRA radio bearer combination 2: SRB1 and SRB2 for DCCH + 1xAM DRB + 1xUM DRB

    f_GenericProcedure_12_1_1(eutra_Cell1, 1, 1);

    // E-UTRA radio bearer combination 4: SRB1 and SRB2 for DCCH + 2xAM DRB + 1xUM DRB

    f_GenericProcedure_12_1_1(eutra_Cell1, 2, 1);

    // E-UTRA radio bearer combination 7: SRB1 and SRB2 for DCCH + 3xAM DRB + 1xUM DRB

    f_GenericProcedure_12_1_1(eutra_Cell1, 3, 1);

    // E-UTRA radio bearer combination 10: SRB1 and SRB2 for DCCH + 4xAM DRB + 1xUM DRB

    f_GenericProcedure_12_1_1(eutra_Cell1, 4, 1);

    f_EUTRA_TestBody_Set(false);

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Post Amble

    //////////////////////////////////////////////////////////////////////////////////

    //Switch off UE and expect a DETACH REQUEST message

    f_EUTRA_Postamble (eutra_Cell1, E3_TESTMODE);

  };


After:

	  function f_TC_12_2_2_EUTRA ( ) runs on EUTRA_PTC

  { /* DRB/12.2

       12.2.2 : Data transfer of E-UTRA radio bearer combinations 2,4,7 and 10

    */

    ///////////////////////////////////////////////////////////////////////////////////

    //      PREAMBLE

    //////////////////////////////////////////////////////////////////////////////////

    var boolean v_TC_12_2_2 := true;                  

    f_EUTRA_Init (c1);

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    // Bring UE to state 3

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true);

    // Test body starts

    // E-UTRA radio bearer combination 2: SRB1 and SRB2 for DCCH + 1xAM DRB + 1xUM DRB

    f_GenericProcedure_12_1_1(eutra_Cell1, 1, 1, v_TC_12_2_2);

    // E-UTRA radio bearer combination 4: SRB1 and SRB2 for DCCH + 2xAM DRB + 1xUM DRB

    f_GenericProcedure_12_1_1(eutra_Cell1, 2, 1, v_TC_12_2_2);

    // E-UTRA radio bearer combination 7: SRB1 and SRB2 for DCCH + 3xAM DRB + 1xUM DRB

    f_GenericProcedure_12_1_1(eutra_Cell1, 3, 1, v_TC_12_2_2);

    // E-UTRA radio bearer combination 10: SRB1 and SRB2 for DCCH + 4xAM DRB + 1xUM DRB

    f_GenericProcedure_12_1_1(eutra_Cell1, 4, 1, v_TC_12_2_2);

    f_EUTRA_TestBody_Set(false);

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Post Amble

    //////////////////////////////////////////////////////////////////////////////////

    //Switch off UE and expect a DETACH REQUEST message

    f_EUTRA_Postamble (eutra_Cell1, E1_IDLE);         

  };
  function f_GenericProcedure_12_1_1 (CellId_Type p_CellId,

                                      integer p_NoOfAddDRBs_AM,

                                      integer p_NoOfAddDRBs_UM,

                                      boolean p_TC_12_2_2 := false) runs on EUTRA_PTC                 

  { //@sic R5s100100

    ……………

    ……………
    // SS configuration for approprite DRB's

    f_EUTRA_SS_DRB_AddConfig(p_CellId, p_NoOfAddDRBs_AM-1, p_NoOfAddDRBs_UM);

    // Generate TFTFilters and QOS

    for (n := 0; n < (p_NoOfAddDRBs_AM-1 + p_NoOfAddDRBs_UM); n := n + 1)

      {

        v_QoSList[n] := cs_508_EPS_QoS_Dedicated_1;

        select (n) {

          case (0) {

            v_Tft_List[n] := cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId);

          }

          case (1) {

            v_Tft_List[n] := cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId2); //R5s100274

          }

        //@sic R5s100116

        case (2) {

            v_Tft_List[n] := cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId3);  //R5s100274

          }

          case (3) {

            v_Tft_List[n] := cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId4); 

          }

          case (4) {

            v_Tft_List[n] := cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId5);  //R5s100274

          }

          case (5) {

            v_Tft_List[n] := cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId6); 

          }

          case (6) {

            v_Tft_List[n] := cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId7);  //R5s100274

          }

          case else {

            FatalError(__FILE__, __LINE__, "not supported total number of bearers");

            /// dummy statement added for satisfying the compiler

          }

        } // end of select

      }// end of for

    // Num AM bearers 1 less as Default bearer is not included which is configured by default.

    if (p_TC_12_2_2 == true)        

    {

       f_EUTRA_LoopBackActivation_State4_Generic_1222(p_CellId, cs_UE_TestLoopModeA_LB_Setup_NoScaling, p_NoOfAddDRBs_AM-1, p_NoOfAddDRBs_UM, v_QoSList, v_Tft_List);

    }

    else

    {

       f_EUTRA_LoopBackActivation_State4_Generic(p_CellId, cs_UE_TestLoopModeA_LB_Setup_NoScaling, p_NoOfAddDRBs_AM-1, p_NoOfAddDRBs_UM, v_QoSList, v_Tft_List);

    }
 ……..

 ...…...
  };
  function f_EUTRA_LoopBackActivation_State4_Generic_1222(CellId_Type p_CellId,

                                                     template (value) UE_TestLoopModeA_LB_Setup_Type p_UE_TestLoopModeA_LB_Setup,

                                                     integer p_NoOfAddDRBs_AM,

                                                     integer p_NoOfAddDRBs_UM,

                                                     template (omit) EPS_QoSList_Type p_QoS,

                                                     template (omit) Tft_List_Type p_Tft ) runs on EUTRA_PTC

  {

    fl_EUTRA_GenericRbEst_1222 (p_CellId, p_NoOfAddDRBs_AM, p_NoOfAddDRBs_UM, p_QoS, p_Tft);      // bring UE to State 3

    f_EUTRA_CloseUE_TestLoopModeA(p_CellId, p_UE_TestLoopModeA_LB_Setup);             // Step 1-2

  };

  function fl_EUTRA_GenericRbEst_1222 ( CellId_Type p_CellId,

                                   integer p_NoOfAddDRBs_AM,

                                   integer p_NoOfAddDRBs_UM,

                                   template (omit) EPS_QoSList_Type p_QoS,

                                   template (omit) Tft_List_Type p_Tft ) runs on EUTRA_PTC

  {

    var template (omit) DRB_ToAddModList v_DRB_ToAddModList := omit;                // => apply default setiing acc. 36.508 cl. 4.8.2.1.7 */

    var template (omit) MAC_MainConfig_Type v_MAC_MainConfig := omit;               // => apply default setting acc. 36.508 cl. 4.8.2.1.5 */

    var template (omit) PhysicalConfigDedicated v_PhysicalConfigDedicated := omit;  // => apply default setting acc. 36.508 cl. 4.8.2.1.6 */

    f_EUTRA_GenericRbEst_Common_1222(p_CellId,

                                p_NoOfAddDRBs_AM,

                                p_NoOfAddDRBs_UM,

                                v_DRB_ToAddModList,

                                v_MAC_MainConfig,

                                v_PhysicalConfigDedicated,

                                p_QoS,

                                p_Tft);

  }

  function f_EUTRA_GenericRbEst_Common_1222 ( CellId_Type p_CellId,

                                         integer p_NoOfAddDRBs_AM,

                                         integer p_NoOfAddDRBs_UM,

                                         template (omit) DRB_ToAddModList p_DRB_ToAddModList,

                                         template (omit) MAC_MainConfig_Type p_MAC_MainConfig,

                                         template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated,

                                         template (omit) EPS_QoSList_Type p_QoS,

                                         template (omit) Tft_List_Type p_Tft,

                                         boolean p_PDCP_StatusReportRequired := true )

    runs on EUTRA_PTC

  {

    var M_TMSI_Type v_M_Tmsi_Def := tsc_M_TMSI1;

    var GutiParameters_Type v_GutiParameters := f_EUTRA_CellInfo_GetGuti ( p_CellId );

    /* Step 2 */

    f_EUTRA_UE_Page ( p_CellId,

                    cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParameters.MME_Code, v_M_Tmsi_Def )) );

    // Steps 3 - 10

    f_EUTRA_ServiceRequestAndActivate_SRB2_DRB_1222 ( p_CellId,

                                                 p_NoOfAddDRBs_AM,

                                                 p_NoOfAddDRBs_UM,

                                                 p_DRB_ToAddModList,

                                                 p_MAC_MainConfig,

                                                 p_PhysicalConfigDedicated,

                                                 p_QoS,

                                                 p_Tft,

                                                 p_PDCP_StatusReportRequired);

  }

  function f_EUTRA_ServiceRequestAndActivate_SRB2_DRB_1222(CellId_Type p_CellId,

                                                      UInt_Type p_NoOfAddDRBs_AM,

                                                      UInt_Type p_NoOfAddDRBs_UM,

                                                      template (omit) DRB_ToAddModList p_DRB_ToAddModList,

                                                      template (omit) MAC_MainConfig_Type p_MAC_MainConfig,

                                                      template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated,

                                                      template (omit) EPS_QoSList_Type p_QoS,

                                                      template (omit) Tft_List_Type p_Tft,

                                                      boolean p_PDCP_StatusReportRequired := true ) runs on EUTRA_PTC

  {

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_MSG_Indication_Type v_NAS_Ind;

    var NAS_KsiValue v_KsiValue;

    // =====================================================

    // step 3-10 of procedure in 36.508 cl. 4.5.3.3

    // Step 3 - 5 of Generic RB establishment procedure

    //Check the Service Request message

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get ( );

    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas ( p_CellId,

                                                  v_RRC_TI,

                                                  cr_EstablishmentCause_mt_Access,

                                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                    cr_508_SERVICE_REQUEST ( v_KsiValue )));

    /* Step 6 - 10 */

    f_EUTRA_Activate_SRB2_DRB_1222 ( p_CellId,

                                 p_NoOfAddDRBs_AM,

                                 p_NoOfAddDRBs_UM,

                                 p_DRB_ToAddModList,

                                 p_MAC_MainConfig,

                                 p_PhysicalConfigDedicated,

                                 p_QoS,

                                 p_Tft,

                                 v_NAS_Ind,

                                 p_PDCP_StatusReportRequired);

  }

  function f_EUTRA_Activate_SRB2_DRB_1222(CellId_Type p_CellId,

                                      UInt_Type p_NoOfAddDRBs_AM,

                                      UInt_Type p_NoOfAddDRBs_UM,

                                      template (omit) DRB_ToAddModList p_DRB_ToAddModList,

                                      template (omit) MAC_MainConfig_Type p_MAC_MainConfig,

                                      template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated,

                                      template (omit) EPS_QoSList_Type p_QoS,

                                      template (omit) Tft_List_Type p_Tft,

                                      NAS_MSG_Indication_Type p_NAS_Ind,

                                      boolean p_PDCP_StatusReportRequired := true) runs on EUTRA_PTC

  {

    …………….
    …………….

    // -----------------------------------------------------

    // initialise v_DRB_ToAddModList

    if (isvalue(p_DRB_ToAddModList)) {                  // explicit setting for DRB_ToAddMod

      if (lengthof(p_DRB_ToAddModList) != v_NoOfAddDRBs+1) {

        FatalError (__FILE__, __LINE__, "invalid no. of DRB_ToAddMod");

      }

      v_DRB_ToAddModList := p_DRB_ToAddModList;

    } else {                                            // use default configuration

      v_DRB_Id := tsc_DRB1;                             // default bearer

      for (n := 0; n < (p_NoOfAddDRBs_AM + 1); n := n + 1) {                // default bearer included in the list (=> p_NoOfAddDRBs_AM + 1)

        if (p_PDCP_StatusReportRequired == true)

        {

          v_DRB_ToAddModList[n] := cs_508_DRB_ToAddMod_DEFAULT_AM (v_DRB_Id);

        }

        else

        {

          v_DRB_ToAddModList[n] := cs_DRB_ToAddMod_NoPDCP_StatusReport (v_DRB_Id);

        }

        v_DRB_Id := v_DRB_Id + 1;

      }

      for (m := 0; m < p_NoOfAddDRBs_UM; m := m + 1) {

        v_DRB_ToAddModList[n+m] := cs_508_DRB_ToAddMod_DEFAULT_UM_1222 (v_DRB_Id);             

        v_DRB_Id := v_DRB_Id + 1;

      }

    }

    // -----------------------------------------------------

    // initialise RRCConnectionReconfiguration

    /*Since this is not a L2 test case but still UM bearer is being used so this extra check is removed */

    if ((p_NoOfAddDRBs_UM > 0) /*and f_TestcaseIsL2Testcase()*/) {  /* @sic R5-101177, R5-101178, R5-101050 sic@  

                                                                   @sic R5s100116 sic@ */

      v_TimingAdvancePeriodMode := l2Um;     // => PUCCH sync needs to be reconfigured at the SS side to get short TA period

    }

    if (isvalue(p_MAC_MainConfig)) {

      v_MAC_MainConfig := p_MAC_MainConfig;

    } else {

      if (p_NoOfAddDRBs_UM > 0) {

        if (f_TestcaseIsL2Testcase()) {          // @sic R5-101177, R5-101178, R5-101050 sic@

          v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS_L2;

        } else {

          v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS; // short DRX cycle acc. to note in 36.508 table 4.6.3-16

        }

      } else {

        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxL; // long DRX cycle acc. to note in 36.508 table 4.6.3-16

      }

    }

    /* Since UM bearer is being created along with AM bearer DRX cycle of 200ms is used */

    v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS_L2;           

    ……………..
    ……………...

  }

  template (value) DRB_ToAddMod cs_508_DRB_ToAddMod_DEFAULT_UM_1222 (DRB_Identity p_DRB_Id) :=

    /* According to 36.508 cl. 4.8.2.1.7 Table 4.8.2.1.7-1 (using condition UM) */

    /* @status    APPROVED */

    cs_508_DRB_ToAddMod_Common(p_DRB_Id,

                               cs_508_PDCP_Config_DRB_UM_1222,

                               cs_508_RLC_Config_DRB_UM,

                               cs_508_LogicalChannelConfig_DRB_UM);

  template (value) PDCP_Config cs_508_PDCP_Config_DRB_UM_1222 :=

  { /* Default values according to 36.508 cl. 4.8.2.1.2.1 Table 4.8.2.1.2.1-1 */

    /* @status    APPROVED */

    discardTimer := ms300,                                      

    rlc_AM := omit,

    rlc_UM := {

      pdcp_SN_Size := len12bits

    },

    headerCompression := {

      notUsed := NULL

    }

  };



5 Execution Log Files

5.1 LG Electronics L2000 UE
The LG UE passed this test case on the Anite conformance toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file TC_12_2_2_LG-UE.htm:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5.2 Samsung Tanum UE

The Samsung Tanum UE passed this test case on the Anite conformance toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file TC_12_2_2_Samsung-UE.htm:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

	[1]
	R5s100294: This archive comprises html format execution log file and the TTCN file.

